//get unemployment probabilities from ACS
use "$temp/ACS_profile_data", clear

//nobody is in school. Look at years 2009-2013
drop if empstat == 3 //kick NILF folks
keep if (age==19 & educ_cat == 1) | (age==21 & educ_cat == 2) | (age==23 & educ_cat == 3) //focus on fresh graduates
keep if year>=2008 & year<=2015 //year before and then recession period

//indicator variables
gen emp = (empstat == 1)
gen sc = (educ_cat == 2)
gen coll = (educ_cat == 3)

//regressions
probit emp sc coll [w=perwt] if year == 2008

local cons_base = _b[_cons]
local sc_base = _b[sc]
local coll_base = _b[coll]

forval y = 2009/2013{
	probit emp sc coll [w=perwt] if year == `y'
	local cons_frac_`y' = _b[_cons] / `cons_base'
	local sc_frac_`y' = _b[sc] / `sc_base'
	local coll_frac_`y' = _b[coll] / `coll_base'
}

clear
set obs 5
gen cons = .
gen sc = .
gen coll = .
local counter = 0

forval y = 2009/2013{
	local counter `++counter'
	replace cons = `cons_frac_`y'' in `counter'
	replace sc = `sc_frac_`y'' in `counter'
	replace coll = `coll_frac_`y'' in `counter'
}

export delimited "$dir/model/utilities/rec_emp_shocks.csv", novarn replace

//get vector





//

